some equivalents of the axiom of choice

Johan G. F. Belinfante and Ming Li
2005 November 29

In[1]:= **SetDirectory["l:"]; << goedel75.27a; << tools.m**

```
:Package Title: goedel75.27a            2005 November 27 at 7:00 a.m.

It is now:  2005 Nov 29 at 2:12

Loading Simplification Rules

TOOLS.M                      Revised 2005 October 25

weightlimit = 40
```

summary

In this notebook some new rewrite rules are derived that permit the **GOEDEL** program to recognize some variants of the axiom of choice.

cross-sections of sets whose inverse is a function

The axiom of choice is equivalent to the statement that every set admits a cross-section. Since the class **SELECT** consists of all sets that admit a cross-section, this says:

In[2]:= **equal[V, SELECT]**

Out[2]= axch

Because one can use cross-sections of certain sets to construct cross-sections of others, the axiom of choice is equivalent to various other statements of the form **subclass[x, SELECT]**., where **x** is the clss of sets of some particular type. In particular, it suffices to assume that all inverse functions admit cross-sections:

In[3]:= **subclass[image[INVERSE, FUNS], SELECT]**

Out[3]= axch

The function **INVERSE** here can be replaced by either **IMAGE[SWAP]** or its inverse. No new rule is needed for this variant:

In[4]:= **subclass[image[IMAGE[SWAP], FUNS], SELECT]**

Out[4]= axch

In this section, a rewrite rule will be derived that permits the **GOEDEL** program to also recognize the variant in which **INVERSE** is replaced by **inverse[IMAGE[SWAP]]**. The point is that the existence of a cross-section for a set is not affected by the presence or absence of elements that are not ordered pairs. A general statement to this effect is that **SELECT** is invariant under adding non-pairs:

```
In[5]:= image[inverse[IMAGE[id[cart[V, V]]]], SELECT]

Out[5]= SELECT
```

Lemma.

```
In[6]:= ImageComp[IMAGE[id[cart[V, V]]], inverse[IMAGE[SWAP]], x] // Reverse

Out[6]= image[IMAGE[id[cart[V, V]]], image[inverse[IMAGE[SWAP]], x]] == image[INVERSE, x]

In[7]:= image[IMAGE[id[cart[V, V]]], image[inverse[IMAGE[SWAP]], x_]] := image[INVERSE, x]
```

Theorem.

```
In[8]:= SubstTest[subclass, image[inverse[IMAGE[SWAP]], x],
         image[inverse[IMAGE[id[cart[V, V]]]], y], y → SELECT]

Out[8]= subclass[image[inverse[IMAGE[SWAP]], x], SELECT] == subclass[image[INVERSE, x], SELECT]

In[9]:= subclass[image[inverse[IMAGE[SWAP]], x_], SELECT] := subclass[image[INVERSE, x], SELECT]
```

As a corollary the **GOEDEL** program can now recognize this variant of the axiom of choice:

```
In[10]:= subclass[image[inverse[IMAGE[SWAP]], FUNS], SELECT]

Out[10]= axch
```

Another corollary of a more general nature is the following simplification rule:

```
In[11]:= SubstTest[subclass, x, image[inverse[IMAGE[id[cart[V, V]]]], y], y → SELECT] // Reverse

Out[11]= subclass[image[IMAGE[id[cart[V, V]]], x], SELECT] == subclass[x, SELECT]

In[12]:= subclass[image[IMAGE[id[cart[V, V]]], x_], SELECT] := subclass[x, SELECT]
```

Whether or not the axiom of choice holds, certain relations have cross-sections. For example, **id[fix[x]]** is a cross-section of any (small) reflexive relation **x**. Hence the class of reflexive relations is contained in **SELECT**.

```
In[13]:= subclass[RFX, SELECT]

Out[13]= True
```

Here again, it does no harm to add non-pairs:

```
In[14]:= SubstTest[subclass, image[inverse[IMAGE[id[cart[V, V]]]], RFX],
          image[inverse[IMAGE[id[cart[V, V]]]], y], y → SELECT]

Out[14]= subclass[image[inverse[IMAGE[id[cart[V, V]]]], RFX], SELECT] == True
```

```
In[15]:= subclass[image[inverse[IMAGE[id[cart[V, V]]]], RFX], SELECT] := True
```

Russell's form of axch

Russell's form of the axiom of choice involves pairwise disjoint collections of sets. The class of all such collections is

```
In[16]:= class[x,
          forall[u, v, implies[and[member[u, x], member[v, x]], or[equal[u, v], disjoint[u, v]]]]]

Out[16]= cliques[union[DISJOINT, Id]]
```

A pairwise disjoint collection of sets may or may not hold the empty set. Since one cannot select an element from the empty set, one may simply chose to add a hypothesis banishing the empty set. The following formulation of **axch** results: for any collection of non-empty pairwise disjoint sets, there is a set whose intersection with every member of the collection is a singleton.

```
In[17]:= assert[forall[x, implies[and[not[member[0, x]], member[x, u]],
              exists[y, forall[w, implies[member[w, x], member[intersection[y, w],
                  range[SINGLETON]]]]]]]]] /. u → cliques[union[DISJOINT, Id]]

Out[17]= axch
```

The class of all collections that satisfy the hypothesis is

```
In[19]:= class[x, and[not[member[0, x]], member[x, u]]] /. u -> cliques[union[DISJOINT, Id]]

Out[19]= intersection[cliques[union[DISJOINT, Id]], P[complement[set[0]]]]
```

The class of all collections that satisfy the conclusion is

```
In[18]:= class[x, exists[y,
           forall[w, implies[member[w, x], member[intersection[y, w], range[SINGLETON]]]]]]

Out[18]= domain[UB[image[inverse[CAP], range[SINGLETON]]]]
```

This version of the axiom of choice can thus be written without quantifiers as follows:

```
In[20]:= subclass[intersection[cliques[union[DISJOINT, Id]], P[complement[set[0]]]],
          domain[UB[image[inverse[CAP], range[SINGLETON]]]]]

Out[20]= axch
```

Instead of banishing the empty set in the hypothesis, one could allow the empty set to be part of the collection, provided that the conclusion be modified accordingly:

```
In[21]:= assert[forall[x, implies[member[x, u],
              exists[y, forall[w, implies[and[not[empty[w]], member[w, x]], member[
                  intersection[y, w], range[SINGLETON]]]]]]]]] /. u -> cliques[union[DISJOINT, Id]]

Out[21]= axch
```

The class of all collections that satisfy this modified conclusion is:

```
In[23]:= class[x, exists[y, forall[w, implies[and[not[empty[w]], member[w, x]],
             member[intersection[y, w], range[SINGLETON]]]]]]

Out[23]= image[inverse[IMAGE[id[complement[set[0]]]]],
          domain[UB[image[inverse[CAP], range[SINGLETON]]]]]
```

A variable-free and quantifier-free statement of this variant of Russell's form of the axiom of choice is:

```
In[24]:= subclass[cliques[union[DISJOINT, Id]], image[inverse[IMAGE[id[complement[set[0]]]]],
           domain[UB[image[inverse[CAP], range[SINGLETON]]]]]]

Out[24]= axch
```

If one were to add an **assert** in the process of removing the quantifiers in the conclusion, a slightly more complicated formulation would be obtained:

```
In[25]:= class[x, exists[y, assert[forall[w, implies[and[not[empty[w]], member[w, x]],
              member[intersection[y, w], range[SINGLETON]]]]]]]

Out[25]= image[inverse[IMAGE[id[complement[set[0]]]]],
          fix[composite[inverse[UB[composite[E, inverse[E]]]],
            UB[image[inverse[CAP], range[SINGLETON]]]]]]
```

A new rewrite rule is needed to cope with this complicated-looking fixed-point set. The following simple rewrite rule suffices:

```
In[26]:= subclass[image[inverse[CAP], range[SINGLETON]], composite[E, inverse[E]]] // AssertTest

Out[26]= subclass[image[inverse[CAP], range[SINGLETON]], composite[E, inverse[E]]] == True

In[27]:= subclass[image[inverse[CAP], range[SINGLETON]], composite[E, inverse[E]]] := True
```

With this new rewrite rule in place, the **fix** expression automatically simplifies:

```
In[28]:= fix[composite[inverse[UB[composite[E, inverse[E]]]],
           UB[image[inverse[CAP], range[SINGLETON]]]]]

Out[28]= domain[UB[image[inverse[CAP], range[SINGLETON]]]]
```